
The Tech-X contribution to the NTCC project was completed on 03/31/06.
Below are some of the highlights of the final year.

A TEQ users’ meeting was held at the Sherwood 2005 conference and
a tech-support mail list was created <teq-users@fusion.txcorp.com>. The
stand-alone separatrix module was added to the NTCC repository and is
available on the web.

For the main TEQ module a portable build system was developed (based
on GNU Autotools and similar to the the separatrix build system). Espe-
cially IBM xlf had problems with mixed code (F77 with F90 snippets) in the
same file and approximately 6000 lines of code was rewritten as pure F90.
Circular dependencies between F90 modules were resolved to robustly al-
low correct compilation order. Exception handling was implemented in both
the separatrix and TEQ modules and an user manual was written for TEQ.
Johan Carlsson visited LLNL 05/16/05–05/20/05.

The TEQ module was delivered to Doug McCune for use in pTRANSP.
Tech support for using TEQ for fixed-boundary solves in pTRANSP was
given to Rob Andre. Jim Wiley at IFS became a TEQ user and tech support
was given to him for doing deadstart (free-boundary solve done from scratch)
for the EAST tokamak. Alexei Pankin at Lehigh also became a TEQ user
and some tech support was given to him to make TEQ run on MacOSX.

As the Tech-X active involvement in NTCC is over for now, the TEQ
project was handed off to LLNL. All the source code, including source for
the user manuals, the build scripts and supporting documentation are under
CVS version control at hydra.gat.com. Tech-X will continue to host the mail
list, with Johan Carlsson as unfunded moderator.

The user manuals for the separatrix and TEQ modules are attached as
appendices to this final report.

The interface to the separatrix library is summarized in the “Introduction”
section. A detailed description follows in the “Interface specification” and
commented usage examples can be found in the “Examples” section.

Introduction

CGS units are used for all quantities. The floating-point precision is deter-
mined by the kind parameter rq, which is defined in the file utils.f90. By
default rq is chosen to make real(rq) an 8-byte floating-point type. By
changing line 7 of utils.f90 from rq=r8 to rq=r4, real(rq) becomes a
4-byte floating-point type instead.

The separatrix library takes a numerical MHD equilibrium (solution to
the Grad-Shafranov equation) as input and finds the X- and O-points, the
separatrix and an arbitrary set of flux surfaces. The library is accessed
through the interface defined in the module seplib. The interface is con-
stituted by the four subroutines init sep, exit sep, find separatrix and
find surfaces.

init sep takes 4 arguments: msrf, mls, geom and sep. msrf is the num-
ber of flux surfaces and mls is the number of coordinate points used to define
the separatrix and each of the flux surfaces. The variables geom and sep

are of derived types. geom holds information about the magnetic geome-
try (including the set of msrf flux surfaces) and sep holds the coordinates
of the separatrix. init sep allocates memory for the variables geom and sep.

exit sep takes the same 4 arguments as init sep and it does the oppo-
site: it releases the memory allocated by init sep.

init sep must be called before any other operations on the separatrix
library are performed. exit sep should be called after all other operations
on the separatrix library have been performed.

find separatrix takes 4 arguments: eq, geom, xpoint and sep, where
all are of derived types. eq holds the numerical equilibrium, geom and sep are
as described above, and xpoint holds the coordinates of the X-point. The
primary input to find separatrix is the numerical equilibrium contained

in eq and the primary output is the X-point and separatrix coordinates con-
tained in xpoint and sep, respectively.

find surfaces takes geom as its single argument and fills in the coordi-
nates of the flux surfaces held by geom. It must be called after find separatrix.

Interface specification

The data passed through the interface is encapsulated in the four variables eq,
geom, xpoint and sep, which are all of derived type. The type declarations
are in the seplib module in the file seplib.f90. For eq we have:

! Derived type that describes a numerical equilibrium and its meta data

type :: eq_type

! R and Z are the gridpoints where psi is given.

real(rq), dimension(:), pointer :: R, Z

! psi is the stream function (poloidal flux / 2pi).

real(rq), dimension(:), pointer :: psi

! Coordinates of magnetic axis, psi at magnetic axis.

real(rq) :: raxis, zaxis, psix

! Limiter point coordinates and psi.

real(rq) :: rlim, zlim, psil

! nsym = 1 for up/down symmetric equilibrium, 2 for asymmetric.

integer :: nsym

end type eq_type

eq is predominantly an input variable and must be initialized before find separatrix

and find surfaces are called. Our demonstration driver initializes eq from
an EQDSK g-file:

subroutine read_eqdsk(ioun, goun, eq)

! We need the derived type for the equilibrium (eq_type) from seplib

use seplib, only : rq, eq_type

implicit none

type(eq_type) :: eq

integer, intent(in) :: ioun

character*(*), intent(in) :: goun

real(rq), dimension(:), allocatable :: dummyd

real(rq) :: dr, dz, dummy, ro, rdim, zdim

integer :: i, jm, km, j, k

print *, ioun, goun

open(unit = ioun, file = goun)

read(ioun, ’(52x,2i4)’) jm, km

allocate(eq%psi(jm * km), eq%R(jm), eq%Z(km), dummyd(jm))

read(ioun, ’(5e16.9)’) rdim, zdim, dummy, ro, dummy

read(ioun, ’(5e16.9)’) dummy, dummy, dummy, dummy, dummy

read(ioun, ’(5e16.9)’) dummy, dummy, dummy, dummy, dummy

read(ioun, ’(5e16.9)’) dummy, dummy, dummy, dummy, dummy

read(ioun, ’(5e16.9)’) (dummyd(i), i = 1, jm)

read(ioun, ’(5e16.9)’) (dummyd(i), i = 1, jm)

read(ioun, ’(5e16.9)’) (dummyd(i), i = 1, jm)

read(ioun, ’(5e16.9)’) (dummyd(i), i = 1, jm)

read(ioun, ’(5e16.9)’) (eq%psi(i), i = 1, jm * km)

close(ioun)

deallocate(dummyd)

dr = rdim / (jm - 1)

dz = zdim / (km - 1)

eq%R = (/(i, i = 0, jm - 1)/) * dr + ro

eq%Z = (/(i, i = 0, km - 1)/) * dz - zdim / 2_rq

! Convert to CGS

eq%psi = eq%psi * 1.0e+08_rq

eq%R = eq%R * 100_rq

eq%Z = eq%Z * 100_rq

end subroutine read_eqdsk

For codes that do not load EQDSK g-files, the initialization of eq will obvi-
ously be implemented differently.

The only components of eq that are not strictly input variables are
eq%raxis, eq%zaxis and eq%psix. If eq%raxis is non-zero, the coordi-
nate (eq%raxis, eq%zaxis) is used as an initial guess for the axis loca-
tion. If eq%raxis is set to zero, the geometric axis is used instead. In
either case eq%raxis and eq%zaxis are overwritten with the values found by
find separatrix.

geom has the following type declaration:

! Auxilliary types for geom_type

type :: R_type

! R coordinate, partial derivatives of R WRT theta and psi, respectively,

! and poloidal magnetic field

real(rq), dimension(:), pointer :: coord, theta, psi, Bpol

end type R_type

type :: Z_type

! Z coordinate, partial derivatives of Z WRT theta and psi, respectively,

! and Jacobian

real(rq), dimension(:), pointer :: coord, theta, psi, J

end type Z_type

! Relates cylinder coordinates in configuration space to flux coordinates

type :: geom_type

type(R_type), dimension(:), pointer :: R

type(Z_type), dimension(:), pointer :: Z

! Normalized psi values at the flux surfaces

real(rq), dimension(:), pointer :: psibar

! Poloidal angles along each flux surface where R, Z are calculated

real(rq), dimension(:), pointer :: tswpts

! if wvert is inputted as non-zero then its outputted value is the

! elongation on axis and the bi-polar coordinate used for the flux

! geometry is stretched by that amount:

! z=zaxis+wvert*rho(psi,theta)*cos(theta)

real(rq) :: wvert

end type geom_type

geom is both an input and an output variable. The input components are
geom%psibar and geom%tswpts. In the demonstration driver they are ini-
tialized like this:

! Note psibar coordinate is stretched at the axis and the edge

geom%psibar=(/(i, i = 0, msrf - 1)/) / (msrf - 1.0)

geom%psibar=(sin(geom%psibar * pi2))**2

! Note tswpts must always be uniform (pi2 = pi / 2)

geom%tswpts=(/(4.0 * pi2 * i, i = 0, mls - 1)/) / (mls - 1.0)

geom%psibar is used to determine how the msrf flux surfaces should be dis-
tributed. Similarly, tswpts determines the mls poloidal angles at which the
flux surface and separatrix (R,Z) coordinates are calculated.

The output components are geom%R and geom%Z, two arrays with msrf

elements each. The elements of geom%R and geom%Z, one per flux surface,
are themselves arrays with each of their mls elements corresponding to a
poloidal point along the flux surface. In addition to the (R,Z) coordinates,
these elements contain (∂R/∂θ, ∂R/∂ψ,Bp) and (∂Z/∂θ, ∂Z/∂ψ, J) (where
J is the Jacobian), respectively. The values of geom%R and geom%Z are cal-
culated and filled in by find surfaces. Note that find separatrix must
be called before find surfaces.

geom%wvert is most likely only of interest to expert users, who are un-
doubtedly capable of parsing the source code to understand how geom%wvert

is properly used.

xpoint holds the location of the X-point and associated information.

! Derived type that describes an X-point

type :: x_point_type

! X-point coordinates and psi.

real(rq) :: rxpt, zxpt, pxpt

! rxpr(2) and zxpr(2) define the box in which to search for the X-point

! If rxpr(1) = 0 the search is not limited

real(rq), dimension(2) :: rxpr, zxpr

! If ixpt=2: uses X-point to define plasma boundary

! If ixpt=1: uses X-point or limiter point, whichever is active

! If ixpt=0: does not look for X-point and it better not be active

integer :: ixpt

end type x_point_type

xpoint is both an input and an output variable. The input components are
the flag xpoint%ixpt and the size-2 arrays xpoint%rxpt and xpoint%zxpt.
The flag ixpt determines how the plasma boundary is defined. If there is
no limiter, we can use the X-point to define the plasma boundary (ixpt
= 2). If there is a limiter, ixpt = 1 tells the separatrix library to use
the limiter point to define the plasma boundary, unless the X-point is in-
side it. Finally, ixpt = 0 can be used if we are not interested in the X-
point but need to call find separatrix before we can call find surfaces.
xpoint%rxpr and xpoint%zxpr can be used to box in the X-point for particu-
larly difficult equilibria, like SSPX. If xpoint%rxpr(1) = 0.0, xpoint%rxpr
and xpoint%zxpr are ignored; otherwise the search for the X-point is lim-
ited to the box with the corners (rxpr(1), zxpr(1)), (rxpr(1), zxpr(2)),
(rxpr(2), zxpr(2)) and (rxpr(2), zxpr(1)). The output components are
xpoint%rxpt, xpoint%zxpt (coordinates of the X-point) and xpoint%pxpt

(flux at the X-point).

sep is primarily an output variable that holds the coordinates of the
separatrix:

! Derived type that holds the coordinates of the separatrix

type :: sep_type

real(rq), dimension(:), pointer :: R, Z

! theta is the increment in normalized flux away from the separatrix

! at which the plasma boundary is prescribed. That is if theta=0, the

! separatrix is the boundary; if theta=0.01 the 99% flux surface is

! the boundary

real(rq) :: theta

end type sep_type

The input component is sep%theta. For the default value of sep%theta =

0.0, the library calculates the true separatrix. By slightly increasing the
value of sep%theta, we can tell the library to instead find a flux surface
just inside the true separatrix. For sep%theta = 0.01, it finds the 99% flux
surface, etc. This might be useful in the rare cases where find separatrix

fails to find the true separatrix. After a successful call to find separatrix,
the coordinates of the sep%theta-designated surface are in the arrays sep%R
and sep%Z.

The complete interfaces of the four subroutines provided by the separatrix
library are as follows:

subroutine init_sep(msrf, mls, geom, sep)

integer, intent(in) :: msrf, mls

type(geom_type) :: geom

type(sep_type) :: sep

subroutine exit_sep(msrf, mls, geom, sep)

integer, intent(in) :: msrf, mls

type(geom_type) :: geom

type(sep_type) :: sep

subroutine find_separatrix(eq, geom, xpoint, sep)

type(eq_type), intent(inout), target :: eq

type(geom_type), intent(inout) :: geom

type(x_point_type), intent(inout) :: xpoint

type(sep_type), intent(inout), target :: sep

subroutine find_surfaces(geom)

type(geom_type), intent(inout) :: geom

Exception handling

Exception handling was added to the separatrix library in early 2006. The
purpose of exception handling is to return control to the driver when an
exception occurs in the separatrix library, i.e when something goes wrong.
The interface to the exception handling consists of the two subprograms
catch xcpt() and throw xcpt() declared in the module seperr:

module seperr

logical :: exceptions_enabled = .false.

logical :: stop_on_error = .true.

integer :: error_code = 0

character(len = 128) :: error_msg = ’none’

interface

integer function catch_xcpt(arg)

logical, intent(inout) :: arg

end function catch_xcpt

subroutine throw_xcpt(arg)

integer, intent(inout) :: arg

end subroutine throw_xcpt

end interface

end module seperr

Exception handling is enabled by setting stop on error = .false. and
passing the variable exceptions enabled to the function catch xcpt(). If
exception handling is supported on the build platform (Linux with either
lf95, ifort or pgf90 for now), exceptions enabled is set to .true.. When
catch xcpt() is explicitly called it returns the integer zero. When the sub-

routine throw xcpt() is called, the execution of the code will appear

to return from catch xcpt()! The return value of catch xcpt() is the
argument passed to throw xcpt(), unless zero is passed to throw xcpt() in
which case catch xcpt() returns the integer one. Here’s an example from
driver.f90:

! Catch errors that might occur when we initialize the separatrix library

stop_on_error = .false.

ierr = catch_xcpt(exceptions_enabled)

if (ierr /= 0) then

print *, ’The call to init_sep failed, the error message is: ’, &

trim(error_msg), ’. Bailing out...’

stop

end if

! Initialize the separatrix library

call init_sep(msrf, mls, geom, sep)

Here we just print an error message and terminate execution when an excep-
tion occurs, but we could try to handle the exception depending on exactly
what went wrong (determined by the value of ierr), and call init sep again.

8



The exception handling can be made as fine grained as desired (set up for
every single call to the separatrix library), or more coarse grained (set up for
a block of calls). In the remainder of the driver we use a more coarse-grained
approach and group the calls done for each equilibrium. catch xcpt() can be
called an arbitrary number of times, a call to throw xcpt() from anywhere
in the code, no matter how deeply nested the subprogram calls are, will
always return to immediately after the last call to catch xcpt(). The only
exception (no pun intended) is when catch xcpt() is called in a subprogram
(subroutine or function) that returns before throw xcpt() is called, in which
case the behavior is undefined.

Examples

For full details about how to use the separatrix library, look at the source
files seplib.f90 and driver.f90. seplib.f90 contains the seplib module
with declarations of the derived types and the subroutines used to access the
library. driver.f90 contains four commented examples of how to call the
library. Here we will give a summary.

From the file driver.f90:

program driver

! The seplib module defines the interface (subroutines and

! associated derived types) to the separatrix library

use seplib

! Some less important lines snipped out...

! The numerical equilibrium

type(eq_type) :: eq

! The set of flux surfaces

type(geom_type) :: geom

! The X-point

type(x_point_type) :: xpoint

! The separatrix

type(sep_type) :: sep

! Some less important lines snipped out...

! Initialize the separatrix library

call init_sep(msrf, mls, geom, sep)

We first use seplib to get access to the separatrix library. Next we declare
the 4 input/output variables eq, geom, xpoint and sep, and then we call
init sep(msrf, mls, geom, sep) to make the library ready for use. msrf
and mls can be any positive integers, variables or constants. In the driver
they are constants with values 101 and 127, respectively.

Next we set the values of the normalized flux for the set of msrf flux
surfaces and the values of the poloidal angle at the mls coordinate points
along the separatrix and each flux surface:

! Note psibar coordinate is stretched at the axis and the edge

geom%psibar=(/(i, i = 0, msrf - 1)/) / (msrf - 1.0)

geom%psibar=(sin(geom%psibar * pi2))**2

! Note tswpts must always be uniform (pi2 = pi / 2)

geom%tswpts=(/(4.0 * pi2 * i, i = 0, mls - 1)/) / (mls - 1.0)

The next step is to load the numerical equilibrium into the variable eq.
In our demonstration driver this is done by reading in an EQDSK g-file (see
the subroutine read eqdsk for details).

For the up/down asymmetric DIII-D equilibrium used in the first example
there is no limiter, so we need to tell the separatrix library to use the X-point
to define the plasma boundary:

eq%nsym = 2 ! Equilibrium is up/down asymmetric

! We don’t have an estimate of the magnetic axis location

eq%raxis = 0.0; eq%zaxis = 0.0

! There is no limiter

eq%rlim = 0.0; eq%zlim = 0.0

xpoint%ixpt = 2 ! Use X-point for plasma boundary

! First get axis, x_point and separatrix

call find_separatrix(eq, geom, xpoint, sep)

! Some less important lines snipped out...

! Next get all the flux surfaces

call find_surfaces(geom)

Note that find separatrix must be called before find surfaces. A plot
of the output is shown in Fig. 1.

In the next example, with an SSPX equilibrium, we impose a limiter but
still let the X-point define the plasma boundary. Also, we help the separatrix
library find the X-point by narrowing down the search region.

! Set limiter

eq%rlim = 50.222

eq%zlim = -14.7758

xpoint%ixpt = 1 ! Use the limiter or X-point for plasma boundary

! Needs X-point boxed in

xpoint%rxpr(1) = 35.0

xpoint%rxpr(2) = 49.0

xpoint%zxpr(1) = -10.0

xpoint%zxpr(2) = 50.0

! First get axis, x_point and separatrix

call find_separatrix(eq, geom, xpoint, sep)

The output is plotted in Fig. 2.

In the third example, a FIRE equilibrium, we know that there is up/down
symmetry so we pass that information to the library by setting eq%nsym = 1

before we call find separatrix.

Figure 1: DIII-D equilibrium. The dashed line is the separatrix and the
crosses are the O-point (magnetic axis) and X-point.

Figure 2: SSPX equilibrium. The dashed line is the separatrix and the crosses
are the O-point, X-point and the limiter. The solid rectangle is the X-point
search region.

eq%nsym = 1 ! Equilibrium is up/down symmetric

! Set limiter

eq%rlim = 150.0

eq%zlim = 0.0

xpoint%ixpt = 1 ! Use the limiter or X-point for plasma boundary

! First get axis, x_point and separatrix

call find_separatrix(eq, geom, xpoint, sep)

The output is plotted in Fig. 3. The output from the fourth and final exam-
ple, an ITER equilibrium, is plotted in Fig. 4.

Note that exit sep is called after all other operations on the library are
done:

! We’re done, release the allocated memory

call exit_sep(msrf, mls, geom, sep)

end program driver

History

The source code for the separatrix library was extracted from the Corsica
code by Donald Pearlstein, LLNL, and Johan Carlsson, Tech-X Corporation,
and converted from Fortran77/Basis to Fortran90.

Figure 3: FIRE equilibrium. The dashed line is the separatrix and the crosses
are the O-point, X-point and the limiter.

Figure 4: ITER equilibrium. The dashed line is the separatrix and the crosses
are the O-point and the X-point.

Introduction

The TEQ library solves the Grad-Shafranov equation to calculate the free-
boundary (or direct) MHD equilibrium. It can also calculate the fixed-
boundary (or inverse) MHD equilibrium. The TEQ library is quite powerful
and flexible. However, in this first release of the module we will only expose
the basic functionality.

The TEQ module depends on the separatrix module and on LAPACK.
The latest versions can be downloaded from:

• http://fusion.txcorp.com/~johan/teq.tgz

• http://fusion.txcorp.com/~johan/separatrix.tgz

• http://www.netlib.org/lapack/lapack.tgz

Unpack teq.tgz and see the file README for build instructions.

The floating-point precision is determined by the kind parameter rq,
which is defined in the file utils.f90 in the separatrix module. By default
rq is chosen to make real(rq) an 8-byte floating-point type. By chang-
ing line 7 of utils.f90 from rq=r8 to rq=r4, real(rq) becomes a 4-byte
floating-point type instead.

Free-boundary equilibria can currently either be generated from scratch
(“dead start”) or loaded from ASCII TEQ save files. The ability to load
free-boundary equilibria from EQDSK a- and g-files will be added in a future
release. Inverse equilibria can be loaded from TEQ save files.

Interface

To do a dead start (generate a free-boundary equilibrium from scratch)
call the subroutine do dead start("plasma.inp", "coils.inp"), where
the files "plasma.inp" and "coils.inp" contain plasma parameters and coil
configuration, respectively. The file formats are described in Appendix A.

The TEQ module also comes with a set of TEQ save files with pre-
calculated equilibria for the major US machines (DIII-D, C-Mod and NSTX)
and some other ones too (JET, etc.). These files are generated by typing make

test and are used by the demonstration driver program. For most users it
should be sufficient to start with an equilibrium from one of these supplied
files.

A direct solve, for the free-boundary equilibrium, is done by calling the
subroutine find eq. It is the user’s responsibility to first properly initialize
all the input variables, for example by calling load savefile("equilibrium.

sav"), where equilibrium.sav is the name of a TEQ save file.

An inverse solve, for the fixed-boundary equilibrium, is done by calling
the subroutine find inv eq(0). It is again the user’s responsibility to make
sure that a direct (free-boundary) equilibrium has been set up before the
initial inverse solve is attempted. After an initial inverse equilibrium has
been found, or loaded from a TEQ save file, one commonly wants to mod-
ify some quantity and re-solve for the perturbed inverse equilibrium. If the
perturbation is small, the perturbed inverse equilibrium can be found by
find perturbed inv eq. If the perturbation is large it might be necessary
to instead call find inv eq(-1), which is much slower but more robust.

To handle the common situation where the user specifies input using a
flux coordinate different from the one used internally in TEQ, the subrou-
tine translate(psibin,psibar,foo) is provided. Here psibin must be the
normalized poloidal flux of the user-provided quantities and psibar must be
the normalized poloidal flux used by TEQ. The profile foo gets transformed
from the psibin flux coordinate to the psibar one.

The TEQ output profile foo can be transformed back to the user flux
coordinate with the call translate(psibar,psibin,foo). Notice the order
of the arguments.

The user can also interact with the TEQ library by directly setting the
value of the variables in the module teqlib input. The teqlib input mod-
ule variables are described by extensive comments in the file teqinit.f90.

Examples

In this section we’ll walk through and comment select parts of the demon-
stration driver program in driver.f90. The TEQ library is accessed through
the module teqlib. To use the dead-start procedure it is also necessary to
use the module deadstart:

program driver

use teqlib

use deadstart

Next, we load a TEQ save file with an ITER equilibrium:

filename="iter.sav"

call load_savefile(filename)

Calling set verbosity(level) determines how much output will be printed,
more for larger values of level, zero suppresses all output.

call set_verbosity(0)

The first call to find eq is to check that the equilibrium loaded from file
is an actual solution to the Grad-Shafranov equation. In this case it is a
good equilibrium and the direct solve converges in a single iteration. Next,
we reduce the number of flux surfaces used from the default of 101 to 81
and re-solve for the direct equilibrium. We then modify some variables that
control how the solution is calculated and repeatedly re-solve.

call find_eq

msrf = 81

call find_eq

thetac = 0.01 ! move boundary inside separatrix for inverse solver

! or if using qsave (q-profile) in direct solve (ipj=5)

call find_eq ! will reset qsave etc

ipj = 5 ! compute from qsave and psave

ipscl = -1 ! do not scale to fix current or flux

inv_p = 1 ! fixes current to plcm(MA); used if input q-profile

! note that in this case the options are the same

! as for the inverse solver

filter_wp=999 ! smoothes current at edge

call find_eq ! does not reset active save arrays

ipscl = 0 ! fixes plasma current used for all other direct options

! ipscl=1 fixes flux: method scaling

ipj = 1 ! compute from jtsave and psave

call find_eq ! does not reset active save arrays

Next we double the resolution in both the R- and Z-directions and re-solve.
After re-setting the resolution, we set up some parameters that control the
inverse solve before calling find inv eq(0) to calculate the inverse solution.

jm = 2 * jm - 1; km = 2 * km - 1; jmkm = jm * km

liml = -1 ! Uses rn, rx, zn for R-Z grid

call find_eq

jm = (jm + 1) / 2; km = (km + 1) / 2; jmkm = jm * km; liml = -1

call find_eq

nht = 200; epsrk = 1.0e-6

call set_verbosity(2)

inv_p = 0 ! fixes F_bound to fwall

inv_k = 0 ! uses q-profile (qsave)

call find_inv_eq(0) ! use internal contour information.

!call write_savefile("iter_inv.sav")

!inv_p = -1 ! fixes polflux(GS) across plasma to dpsi00 (cgs)

inv_p = 1 ! fixes current, uses plcm (MA)

inv_k = 3 ! uses jparsave (<J.B>/<B.B>

call find_perturbed_inv_eq ! Uses previous solution as starting point

! note inv_k=0 or inv_k=3 are preferred since output profiles

! are equal to input profiles

! inv inv_k=1,2 output profiles must be scaled to input if inv_p=1,-1

Finally the direct (free-boundary) and inverse (fixed-boundary) solutions are
dumped to a TEQ save file.

call write_savefile("driver.sav")

A plot of the direct solution from the file driver.sav, generated by the the
IDL script teqplot.pro, is shown in Fig. 5

It is of course also possible to selectively dump chosen output arrays. The
most commonly used fixed-boundary output arrays are probably the profiles:

Figure 5: The direct (free-boundary) solution calculated by the demonstra-
tion driver.

frsrf (F ), fpsrf (F ′), qsrf (q), jtsrf (Jφ), jparsrf (J‖), prsrf (p) and
pdsrf (p′). Prime denotes the derivative with respect to the stream function
ψ = ψp/2π (poloidal flux over 2π). A more exhaustive description of the
output variables is given by the comments in the file teqinit.f90.

History

The source code for the TEQ library was extracted from the Corsica code
by Donald Pearlstein and Dick Bulmer, LLNL, and Johan Carlsson, Tech-X
Corporation, and converted from Fortran77/Basis to Fortran90.

Appendix A

The dead-start procedure, calling the subroutine do dead start("plasma.inp",

"coils.inp"), requires the two files "plasma.inp" and "coils.inp" to be
set up by hand. The file formats are fairly self-explanatory.

Sample plasma.inp file:

"KSTAR/DN"

Plasma...

2.00 MA plasma current

1.80 m major radius

0.50 m minor radius

0.00 m Zaxis

1.90 95% elongation

0.30 95% triangularity

0.01 m Dsep (DN)

1.00 poloidal beta

0.73 li

-4.00 Wb External flux linkage

Toroidal field...

3.50 T @ R = 1.80 m

Computational grid...

1.00 m Rmin

2.60 m Rmax

-1.50 m Zmin

1.50 m Zmax

33 x 65 No. grid points (Nr x Nz)

Plot Scales...

0.00 m Rmin

4.00 m Rmax

-2.50 m Zmin

2.50 m Zmax

2.00 MA/m^2 Current density for drawing coil cross-sections

Sample coils.inp file:

"KSTAR" PF coil set of 05/01/99 from Kim

14 coils

name Rc [m] Zc [m] DRc [m] DZc [m] n_turn NI_cap B_cap

"PF1U" 0.5610 0.2470 0.2135 0.4764 1 1 1

"PF2U" 0.5610 0.6932 0.2135 0.3808 1 1 1

"PF3U" 0.5610 0.9960 0.2135 0.1896 1 1 1

"PF4U" 0.5610 1.2510 0.2135 0.2852 1 1 1

"PF5U" 1.0850 2.2960 0.3330 0.3808 1 1 1

"PF6U" 3.0900 1.9200 0.1896 0.3808 1 1 1

"PF7U" 3.7300 0.9600 0.1418 0.2852 1 1 1

"PF1L" 0.5610 -0.2470 0.2135 0.4764 1 1 1

"PF2L" 0.5610 -0.6932 0.2135 0.3808 1 1 1

"PF3L" 0.5610 -0.9960 0.2135 0.1896 1 1 1

"PF4L" 0.5610 -1.2510 0.2135 0.2852 1 1 1

"PF5L" 1.0850 -2.2960 0.3330 0.3808 1 1 1

"PF6L" 3.0900 -1.9200 0.1896 0.3808 1 1 1

"PF7L" 3.7300 -0.9600 0.1418 0.2852 1 1 1

